The ninth in a series of tutorials for the beginner to intermediate machine language programmer

Machine Language Made BASIC

Part IX: Let There Be Music

By William P. Nee

The Color Computer is adept at producing musical sounds. Complicated and expensive hardware can replicate almost any musical instrument and play over many octaves with several voices. This month we'll explore the SOUND and PLAY commands and execute them from machine language programs. In a later article, we'll learn how to play music with up to six voices (notes) at one time. However, for right now, let's stick to one note at a time.

Let's start with the SOUND command. To use this command in BASIC, you need to enter a note (1 to 255), followed by the duration of play. Table 1 gives each note and its corresponding number. Middle C is C4, with a value of 89. These values can only approximate the note's actual frequency, but they will produce a good sound.

Notes	Octave				
	3	4	5	6	7
C	---	89*	176	218	239
C#/D-	---	99	180	221	241
D	---	108	185	223	242
D#/E-	---	117	189	225	243
E	---	125	193	227	244
F	5	133	197	229	---
F#/G-	19	140	200	231	---
G	32	147	204	232	---
G#/A-	45	153	207	234	---
A	58	159	210	236	---
A#/B-	69	165	213	237	---
B	78	170	216	238	---

*Middle C

Table 1: SOUND Notes

Load Register A with the desired sound and Register B with the duration; store A in Location $8C and execute the SOUND command at Address $A951. Please note: You'll lose anything stored in registers A and X. Routine 1 plays every note from 1 to 255. A sine-wave table of 36 notes (used in cassette programs) starts at Address $A85C. If you want to play these notes, try Routine 2.

To play a tune with notes of different durations, make up your own note table of two bytes for each note and its duration. Load Register A with the note and Register B with its duration, then play the note. Decrease the note counter and continue until out of notes. Playing notes can also be integrated with your visual display, but that will slow down the tempo.

The PLAY command is more complicated and requires more set-up. It uses the following locations:

Location	Description
$D8	Number of notes, pauses, etc.
$DE	Octave (0 to 4)
$DF/E0	Volume
$E1	Note length
$E2	Tempo
$E5	Number of dots after length

Each note is numbered from one to 12 since there are 12 half-steps in an

---

Bill Nee bucked the "snowbird" trend by retiring to Wisconsin from a banking career in Florida. He spends the long, cold winters writing programs for his CoCo.

```
            ORG   $3000
SOUND       LDD   #$0101       NOTE = 1, DURATION = 1
AGAIN       STA   $8C
            PSHS  A            SAVE NOTE
            JSR   $A951        SOUND NOTE
            PULS  A            GET NOTE
            INCA  NEXT NOTE
            CMPA  #$FF         TOP NOTE?
            BLO   AGAIN        IF LOWER, AGAIN
            SWI
```

Routine 1: Playing the Notes

```
            ORG   $3000
START       LDY   #$A85C       ADDRESS OF NOTE TABLE
            LDB   #1           NOTE DURATION
            LDA   #36          NUMBER OF NOTES TO PLAY
            STA   NOTES        SAVE IT
LOOP        LDA   ,Y+          GET A NOTE
            STA   $8C
            JSR   $A951        PLAY THE NOTE
            DEC   NOTES        ONE LESS TO PLAY
            BNE   LOOP         IF NOT OUT OF NOTES, BACK TO LOOP
DONE        SWI
NOTES       RMB   1
            END   START
```

Routine 2: Notes From a Sine-Wave Table

Note	Location				
	O0	O1	O2	O3	O4
	$9C62	$9C7A	$9C92	$9C9E	$9CAA
C	#$1A8	#$0D3	#$A6	#$51	#$26
C#	#$190	#$0C7	#$9C	#$4C	#$23
D	#$17A	#$0BB	#$93	#$47	#$21
D#	#$164	#$0B1	#$8B	#$43	#$1F
E	#$150	#$0A6	#$83	#$3F	#$1D
F	#$13D	#$09D	#$7B	#$3B	#$1B
F#	#$12B	#$094	#$74	#$37	#$19
G	#$11A	#$08B	#$6D	#$34	#$18
G#	#$10A	#$083	#$67	#$31	#$16
A	#$0FB	#$07C	#$61	#$2E	#$14
A#	#$0ED	#$075	#$5B	#$2B	#$13
B	#$0DF	#$06E	#$56	#$28	#$12

Table 2: PLAY Delay Cycles

Volume:	$DF/E0:	Volume:	$DF/E0:
V31	#$FA02	V30	#$F606
V29	#$F20A	V28	#$EE0E
V27	#$EA12	V26	#$E616
V25	#$E21A	V24	#$DE1E
V23	#$DA22	V22	#$D626
V21	#$D22A	V20	#$CE2E
V19	#$CA32	V18	#$C636
V17	#$C23A	V16	#$BE3E
V15	#$BA42	V14	#$B646
V13	#$B24A	V12	#$AE4E
V11	#$AA52	V10	#$A656
V9	#$A25A	V8	#$9E5E
V7	#$9A62	V6	#$9666
V5	#$926A	V4	#$8E6E
V3	#$8A72	V2	#$8676
V1	#$827A	V0	#$7E7E

Table 3: PLAY Volume

octave (C natural, C sharp/D flat, D natural, D sharp/E flat, E natural, F natural, F sharp/G flat, G natural, G sharp/A flat, A natural, A sharp/B flat and B natural).

There are five available octaves, but the computer subtracts one from the octave number, giving us octaves 0 to 4. The frequency table for the five octaves begins at $9C62. (See Table 2). The volume is a two-byte number corresponding to V31, V0 (Volume 0) in BASIC. Table 3 gives the BASIC volume and the corresponding number that goes in locations $DF and $E0. The first number is 126 plus four times the volume; the second number is 126 minus four times the volume.

The length can be any value between 1 and 255. Adding a dot after the value increases the value by one half. The common notes and their lengths are as follows:

Note	Length
Whole	L1
Half	L2
Quarter	L4
Quarter Triplet	L6
Eighth	L8
Eighth Triplet	L12
Sixteenth	L16
Sixteenth Triplet	L24
Thirty-second	L32
Thirty-second Triplet	L48

The tempo can be any number between one and 255. The computer defaults to a tempo of two at power-up. Use the same length values as above for

a pause or rest. A pause actually plays a note, but at V0. The number of notes and pauses to be played goes into Location $D8, and the location of your note table goes into $D9/DA. Since $D8 is a one-byte location, you cannot have a note table of more than 255 notes. Any more will require a second note table.

The PLAY command goes from Address $9A22 to $9CB5, taking up 660 bytes — quite a routine. Fortunately there is a way to get around entering individual volumes, tempos, notes and lengths. This method involves loading your note table location into Location $A6/A7 (the current pointer location) and creating a note table using the *EDTASM* FCC (Form Constant Character) op code. The note table must start and end with quotes, just as the PLAY command would. Use all the PLAY command notations, such as notes A to G, octaves 1 to 5, lengths 1 to 255, tempo, pause, etc. After using the PLAY routine at $9A22, reload Location $A6/A7 with its original value. This routine *cannot* be executed from *ZBUG*; you must be in BASIC. Remember: Once you go to BASIC, your machine language program buffer is lost. It's still in *ZBUG*, but the source code has vanished. Try Routine 3; notice that NTAB1 starts at $301D and NTAB2 starts at $3056. Jot down those locations if you want to correct or change any notes. While in *ZBUG*, use the A mode to find that the last byte used ('') is at $30CE. If you want to add any more notes, they would have to start after that location.

When there are no errors in the source code, enter Q to return to BASIC, then enter EXEC &H3000 to play the music. If you decide to slow down the tempo, enter EXEC &HC000 to return to *EDTASM*, then Z to get to *ZBUG*. Since NTAB1 contains the tempo, type A for the ASCII mode and 301D/ to get to NTAB1. Continue pressing the down-arrow key until you get past the T. The next byte contains the original tempo of 4. Enter 3E to change the tempo and return to the edit mode, then press Q to return to BASIC. Type EXEC &H3000 again, and the same music plays — but at a slower speed.

When you first power up, the subroutine at Address $829C sets the octave (O3), the volume (V15), the length of the note (L4) and the note's tone (T2). Unless you're going to change one of these, you don't need to enter them. Regardless of what any manual states, the scale goes from C to B in each octave.

```
          ORG   $3000
START     LDX   $A6          SAVE CURRENT POINTER LOCATION
          PSHS  X
          LDX   #NTAB1       REPLACE IT WITH NOTE TABLE 1 LOCATION
          STX   $A6
          JSR   $9A22        PLAY NOTE TABLE 1
          LDX   #NTAB2       REPLACE IT WITH NOTE TABLE 2 LOCATION
          STX   $A6
          JSR   $9A22        PLAY NOTE TABLE 2
          PULS  X            GET OLD POINTER AND PUT IT BACK
          STX   $A6
          RTS                RETURN TO BASIC
NTAB1     FCC   /"T4V3003L4CP100C/
          FCC   /GP100GAP100AGP100G/
          FCC   /FP100FEP100E/
          FCC   /DP100L8.DL16EL2C"/
NTAB2     FCC   /"L4GP100GFP100FEP100E/
          FCC   /DP100DGP100GFP100F/
          FCC   /EL100FEDL8.EL10F/
          FCC   /L4EDCP100CGP100GAP100A/
          FCC   /GP100GFP100FEP100E/
          FCC   /DL100EDCL8.DL16EL2C"/
          END   START
```

Routine 3: Replacing Locations

The Listing: MLNOTES

```
3000                  0010Ø          ORG     $3000
3000 9E   A6          0011Ø START    LDX     $A6
3002 34   10          0012Ø          PSHS    X
3004 8E   3019        0013Ø          LDX     #NTAB1
3007 9F   A6          0014Ø          STX     $A6
3009 BD   9A22        0015Ø          JSR     $9A22
300C 8E   3058        0016Ø          LDX     #NTAB2
300F 9F   A6          0017Ø          STX     $A6
3011 BD   9A22        0018Ø          JSR     $9A22
3014 35   10          0019Ø          PULS    X
3016 9F   A6          0020Ø          STX     $A6
3018 39               0021Ø          RTS
3019      22          0022Ø NTAB1    FCC     /"T4V3Ø3L4CP1ØØC/
3029      47          0023Ø          FCC     /GP1ØØGAP1ØØAGP1ØØG/
303B      46          0024Ø          FCC     /FP1ØØFEP1ØØE/
3047      44          0025Ø          FCC     /DP1ØØL8.DL16EL2C"/
3058      22          0026Ø NTAB2    FCC     /"L4GP1ØØGFP1ØØFEP1ØØE/
306D      44          0027Ø          FCC     /DP1ØØDGP1ØØGFP1ØØF/
307F      45          0028Ø          FCC     /EL1ØØFEDL8.EL1ØF/
308F      4C          0029Ø          FCC     /L4EDCP1ØØCGP1ØØGAP1ØØA/
30A5      47          0030Ø          FCC     /GP1ØØGFP1ØØFEP1ØØE/
30B7      44          0031Ø          FCC     /DL1ØØEDCL8.DL16EL2C"/
          3000        0032Ø          END     START

ØØØØØ TOTAL ERRORS
```

My program offers a simple tune, but don't stop with my tune. The musical possibilities are endless. You only need a CoCo, some imagination and some patience.

Questions or comments about this tutorial may be directed to the author at Route 2, Box 216C, Mason, WI 54856-9302. Please enclose an SASE when requesting a reply.)